
Lab 4: Beginning to Build the 32-bit ALU

A. Netlist


The following is the netlist of the adder/subtractor/comparator circuitry.  The comparator is only used if the process is subtraction.

// Joseph Troy Espiritu

// Lloyd Vincent Santos

// CS 152 A (Monday Lab)

******************

*** 32 BIT ALU ***

******************

.include "stdcell.jsim" // to use subcircuits in stdcell.jsim

** INPUT VALUES **

Wtest1 a[31:0] nrz(0v,5.0v,50ns,0ns,.1ns,.1ns) 0x00000002 0x80000000 0x00000003 0x7FFFFFFF 0x00000003

Wtest2 b[31:0] nrz(0v,5.0v,50ns,0ns,.1ns,.1ns) 0xFFFFFFFD 0x00000002 0x00000005 0xFFFFFFFE 0x00000003

.connect cmp[31:1] 0 // cmp31 tp cmp1 is just 0

Xadd   add   constant0 // value for add is 0

Xminus minus constant1 // value for minus is 0

** The following are constants for the comparator .

** These are added to facilitate testing the design.

// case for EQUAL

Xequal1 equal1 constant1

Xlessthan1 lessthan1 constant0

Xlessthaneq1 lessthaneq1 constant0

// case for LESS THAN

Xequal2 equal2 constant0

Xlessthan2 lessthan2 constant1

Xlessthaneq2 lessthaneq2 constant0

// case for LESS THAN EQUAL

Xequal13 equal3 constant0

Xlessthan3 lessthan3 constant0

Xlessthaneq3 lessthaneq3 constant1

** INSTANTIATION OF THE 32BITADDER **

X32bit a[31:0] b[31:0] s[31:0] minus v z n 32BITADDER

** SUBCIRCUIT: Comparator Subcircuit

// This is the subcircuit of the comparator.

// op1 op2 op3 are the inputs for the cases of comparison (equal, less than, less than equal)

// n z v are also inputs (if negative, all zeros and overflow)

// cmp0 is the output

.subckt COMPARATOR op1 op2 op3 n z v cmp0

Xand1 op1 z a1 and2

Xxor1 n v x1 xor2

Xand2 op2 x1 a2 and2

Xor1 z x1 o1 or2

Xand3 o1 op3 a3 and2

Xor3 a1 a2 a3 cmp0 or3

.ends

** SUBCIRCUIT: Full Adder (FA)

// This is the subcircuit for the full adder taken from the previous lab exercise.

// A B Ci are the inputs (A, B, Carry in)

// S and Co are outputs (Sum, Carry out)

.subckt FA A B Ci S Co

Xxor1 A B out1 xor2

Xxor2 out1 Ci S xor2

Xand1 A B out2 and2

Xand2 A Ci out3 and2

Xand3 Ci B out4 and2

Xxor3 out2 out3 out4 Co or3

.ends

** SUBCIRCUIT: ADDER (Four bit)

// This is the subcircuit for the 4bit adder also taken from a previous lab.

// One difference is that a[3:0] is used instead of putting a3 a2 a1 a0

// a[3:0] b[3:0] ci are inputs

// s[3:0] co are outputs

.subckt ADDER a[3:0] b[3:0] s[3:0] co ci

* nodes c0 through c3 are internal to the ADDER module

Xbit0 a0 b0 ci s0 c0 FA

Xbit1 a1 b1 c0 s1 c1 FA

Xbit2 a2 b2 c1 s2 c2 FA

Xbit3 a3 b3 c2 s3 co FA

.ends

** SUBCIRCUIT: 32 XORs (32 XOR gates)

// This is a subcircuit for 32 XOR gates

// This would help in flipping the bits if the operation is subtraction.

// x[31:0] are the inputs

// in would be the second input (usu. 0 if addition; 1 if subtraction)

// y[31:0] are the outputs

.subckt 32XOR x[31:0] in y[31:0]

Xxor31 x31 in y31 xor2

Xxor30 x30 in y30 xor2

Xxor29 x29 in y29 xor2

Xxor28 x28 in y28 xor2

Xxor27 x27 in y27 xor2

Xxor26 x26 in y26 xor2

Xxor25 x25 in y25 xor2

Xxor24 x24 in y24 xor2

Xxor23 x23 in y23 xor2

Xxor22 x22 in y22 xor2

Xxor21 x21 in y21 xor2

Xxor20 x20 in y20 xor2

Xxor19 x19 in y19 xor2

Xxor18 x18 in y18 xor2

Xxor17 x17 in y17 xor2

Xxor16 x16 in y16 xor2

Xxor15 x15 in y15 xor2

Xxor14 x14 in y14 xor2

Xxor13 x13 in y13 xor2

Xxor12 x12 in y12 xor2

Xxor11 x11 in y11 xor2

Xxor10 x10 in y10 xor2

Xxor9  x9  in y9  xor2

Xxor8  x8  in y8  xor2

Xxor7  x7  in y7  xor2

Xxor6  x6  in y6  xor2

Xxor5  x5  in y5  xor2

Xxor4  x4  in y4  xor2

Xxor3  x3  in y3  xor2

Xxor2  x2  in y2  xor2

Xxor1  x1  in y1  xor2

Xxor0  x0  in y0  xor2

.ends

** SUBCIRCUIT: OVERFLOW

// This subcircuit checks if there is an overflow of bits.

// a b s are the inputs

// v is the output

.subckt OVERFLOW a b s v

// This design is taken from the lab document.

Xinva a ai inverter

Xinvb b bi inverter

Xinvs s si inverter

Xand1 a  b  si first and3

Xand2 ai bi s  second and3

Xor first second v or2

.ends

** SUBCIRCUIT: ALLZERO

// This subcircuit checks if all the bits are zero.

// To determine that, each bit in the output of the adder/subtractor is passed on to OR gates.

// The overall output of the or gates is inverted.

// The final output would be zero if there is at least 1 bit that is equal to 1.

// The final output would be one if all bits are zero.

//

// s[31:0] are the inputs (the output of the adder/subtractor)

// z is the output (final output)

.subckt ALLZERO s[31:0] z

Xor1 s31 s30 s29 s28 o1 or4

Xor2 s27 s26 s25 s24 o2 or4

Xor3 s23 s22 s21 s20 o3 or4

Xor4 s19 s18 s17 s16 o4 or4

Xor5 s15 s14 s13 s12 o5 or4

Xor6 s11 s10 s9  s8  o6 or4

Xor7 s7  s6  s5  s4  o7 or4

Xor8 s3  s2  s1  s0  o8 or4

XorA o8  o7  o6  o5  oA or4

XorB o4  o3  o2  o1  oB or4

XorZ oA oB oAB or2

Xinv oAB z inverter

.ends

** SUBCIRCUIT: 32BITADDER 

// This is the circuit of the 32 bit adder (and subtractor).

// This is a ripple carry design utilizing the subcircuits done in the previous lab exercise.

// a[31:0] b[31:0] are the inputs (these are the addends)

// s[31:0] is the output (32bits)

// opo is the operation (0 for add; 1 for minus)

// v z n are also outputs (outputs for the checking if negative, allzeros and overflow)

.subckt 32BITADDER a[31:0] b[31:0] s[31:0] opo v z n

// to help in flipping the bits

X32xor1 b[31:0] opo x[31:0] 32XOR // x[31:0] became the flipped bits if opo is 1

// these are the ripple-carry adder design

X4adder1 a[3:0]   x[3:0]   s[3:0]   co1 opo ADDER

X4adder2 a[7:4]   x[7:4]   s[7:4]   co2 co1 ADDER

X4adder3 a[11:8]  x[11:8]  s[11:8]  co3 co2 ADDER

X4adder4 a[15:12] x[15:12] s[15:12] co4 co3 ADDER

X4adder5 a[19:16] x[19:16] s[19:16] co5 co4 ADDER

X4adder6 a[23:20] x[23:20] s[23:20] co6 co5 ADDER

X4adder7 a[27:24] x[27:24] s[27:24] co7 co6 ADDER

X4adder8 a[31:28] x[31:28] s[31:28] co8 co7 ADDER

// this is the instantiation for the output checkers (overflow, allzero, and negative)

Xover a31 x31 s31 v OVERFLOW

Xallz s[31:0]     z ALLZERO

.connect s31 n

.ends

// These are the COMPARATOR instantiations.

// Only one is required. We put 3 to have easy access in checking and testing. :)

Xcomp1 equal1 lessthan1 lessthaneq1 n z v cmpequal COMPARATOR

Xcomp2 equal2 lessthan2 lessthaneq2 n z v cmplessthan COMPARATOR

Xcomp3 equal3 lessthan3 lessthaneq3 n z v cmplessthaneq COMPARATOR

.tran 120ns

// plots a b and s

.plot a[31:0] 

.plot b[31:0]

.plot s[31:0]

// z n v are connected to these nodes so that it is easy for testing. :D

.connect z allzero_Z

.connect n negative_N

.connect v overflow_V

.plot allzero_Z

.plot negative_N

.plot overflow_V

*just to help in the plotting

.plot SubtractionOnly

.plot cmpequal 

.plot cmplessthan

.plot cmplessthaneq

*just to help in the plotting

.plot nextconnecttoequals

.connect cmp0 cmpequal

.plot cmp[31:0]
B. Circuit Architecture

The circuit used for the adder is the ripple-carry version.  Therefore, the focus was cost optimization.  It is the cheapest to make but it is the slowest.  Each adder unit waits until the preceding adder is finished and set its carry out output.  Unlike the carry look-ahead circuit which seems like each adder works simultaneously, this circuit works in a sequential manner.
C. Architecture Statistics

JSim reported that this circuit has 271 gates.  Its equivalent size is 4972.0 (2.  But this whole circuit has constants for add, minus and OP0-OP3 and three instantiation of the comparator circuit that add up to the number of gates and its size. Those were only added for easy testing and to make the code clearer.  Removing the extra instantiations of the comparator, the circuit would only have 259 gates with a size of 4780.0 (2.

To get the propagation delay (tpd​​) critical path of the circuit, the delay of each gate must be taken into consideration.  Here is a summary of the delays of the gates used in the circuit.
	Gate
	tpd (ns)
	3 input OR
	0.21

	2 input AND
	0.14
	4 input OR
	0.29

	3 input AND
	0.15
	2 input XOR
	0.14

	2 input OR
	0.15
	inverter
	0.02



The full adder circuit is composed of 3 2-input AND and 1 3-input OR gates.  A value is sure to pass through 1 AND and 1 OR gate.  Therefore, it would have a delay of 0.14 + 0.21 = 0.33 ns.  Because the circuit for the 32 bit adder/subtractor is a ripple-carry version, delays are accumulated.  The delay of the 32-bit adder/subtractor would amount to 10.56 ns (32 x 0.33 ns).  Input B would pass through 1 2-input XOR gate that has a delay of 0.14ns for flipping the bits.  Adding it, the total delay for it would be 10.70 ns.

There are other outputs that are part of the path.  The sub-circuit for checking if all the values are zero (Z) and the sub-circuit for checking if an overflow occurs (V).


In sub-circuit Z, it is sure that a value would pass through 2 4-input OR gates, 1 2-input OR gate and an inverter.  The total delay for that part would be 0.75 ns ([2 x 0.29ns] + 0.15 + 0.02).


Sub-circuit V would have an inverter, 1 3-input AND gate and 1 2-input OR gate that would be passed through always.  For this sub-circuit, the propagation delay would be 0.32 ns (0.02 + 0.15 + 0.15),

Because Z and V are not chained together, the higher delay would be taken into consideration for the total propagation delay for the critical path. The propagation delay of the critical path would be:



tpd = 10.70ns + 0.75ns = 11.45 ns.
D. Waveform Plots

The following are tests done with the adder.
Part B Tests: (Add)

(Minus)

Summary:

	A
	
	B
	S (Hex)
	All Zero? (Z)
	Negative? (N)
	Overflow? (V)

	0x00000000
	+
	0x00000000
	0x00000000
	True
	False
	False

	0x55555555
	+
	0x00000000
	0x55555555
	False
	False
	False

	0x00000000
	+
	0x55555555
	0x55555555
	False
	False
	False

	0x55555555
	+
	0x55555555
	0xAAAAAAAA
	False
	True
	True

	0xAAAAAAAA
	+
	0x00000000
	0xAAAAAAAA
	False
	True
	False

	0x00000000
	+
	0xAAAAAAAA
	0xAAAAAAAA
	False
	True
	False

	0xAAAAAAAA
	+
	0xAAAAAAAA
	0x55555554
	False
	False
	True

	0x00000001
	+
	0xFFFFFFFF
	0x00000000
	True
	False
	False

	0x00000000
	-
	0x00000000
	0x00000000
	True
	False
	False

	0x00000000
	-
	0xFFFFFFFF
	0x00000001
	False
	False
	False


Part C Tests (Comparison Tests)

Summary:

	A
	
	B
	S (Hex)
	Z
	N
	V
	A=B
	A<B
	A<=B

	0x00000002
	-
	0xFFFFFFFD
	0x00000005
	False
	False
	False
	False
	False
	False

	0x80000000
	-
	0x00000002
	0x7FFFFFFE
	False
	False
	True
	False
	True
	True

	0x00000003
	-
	0x00000005
	0xFFFFFFFE
	False
	True
	False
	False
	True
	True

	0x7FFFFFFF
	-
	0xFFFFFFFE
	0x80000001
	False
	True
	True
	False
	False
	False

	0x00000003
	-
	0x00000003
	0x00000000
	True 
	False
	False
	True
	False
	True


